
This paper presents a formalization of constructive module theory in the intuitionistic type theory of Coq. We build an abstraction layer on top of matrix encodings, in order to represent finitely presented modules, and obtain clean definitions with short proofs justifying that it forms an abelian category. The goal is to use it as a first step to get certified programs for computing topological invariants, like homology groups and Betti numbers.

Introduction

Homological algebra is the study of linear algebra over rings instead of fields, this means that one considers modules instead of vector spaces. Homological techniques are ubiquitous in many branches of mathematics like algebraic topology, algebraic geometry and number theory. Homology was originally introduced by Henri Poincaré in order to compute topological invariants of spaces [START_REF] Poincaré | Analysis situs[END_REF], which provides means for testing whether two spaces cannot be continuously deformed into one another. This paper presents a formalization1 of constructive module theory in type theory, using the Coq proof assistant [START_REF]Coq development team: The Coq Proof Assistant Reference Manual[END_REF] together with the Small Scale Reflection (SSReflect) extension [START_REF] Gonthier | A Small Scale Reflection Extension for the Coq system[END_REF], which provides a potential core of a library of certified homological algebra.

A previous work, that one of the authors was involved in, studied ways to compute homology groups of vector spaces [START_REF] Heras | Computing persistent homology within Coq/SSReflect[END_REF][START_REF] Heras | Towards a certified computation of homology groups for digital images[END_REF] in Coq. When generalizing this to commutative rings the universal coefficient theorem of homology [START_REF] Hatcher | Algebraic Topology[END_REF] states that most of the homological information of an R-module over a ring R can be computed by only doing computations with elements in Z. This means that if we were only interested in computing homology it would not really be necessary to develop the theory of R-modules in general, but instead do it for Zmodules which are well behaved because any matrix can be put in Smith normal form. However, by developing the theory for general rings it should be possible to implement and reason about other functors like cohomology, Ext and Tor as in the Homalg computer algebra package [START_REF] Barakat | homalg -A Meta-Package for Homological Algebra[END_REF].

In [START_REF] Gonthier | Point-Free, Set-Free concrete linear algebra[END_REF], Georges Gonthier shows that the theory of finite dimensional vector spaces can be elegantly implemented in Coq by using matrices to represent subspaces and morphisms, as opposed to an axiomatic approach. The reason why abstract finite dimensional linear algebra can be concretely represented by matrices is because any vector space has a basis (a finite set of generators with no relations among the generators) and any morphism can be represented by a matrix in this canonical basis. However, for modules over rings this is no longer true: consider the ideal (X, Y ) of k[X, Y ], it is a module generated by X and Y which is not free because XY = Y X. This means that the matrixbased approach cannot be directly applied when formalizing module theory. This is why we restrict our attention to finitely generated modules that are finitely presented, that is, modules with a finite number of generators and a finite number of relations among these. In constructive module theory one usually restricts attention to this class of modules and all algorithms can be described by manipulating the presentation matrices [START_REF] Decker | Computing in Algebraic Geometry: A Quick Start using SINGULAR[END_REF][START_REF] Greuel | A Singular Introduction to Commutative Algebra[END_REF][START_REF] Lombardi | Algèbre commutative, Méthodes constructives: Modules projectifs de type fini[END_REF][START_REF] Mines | A Course in Constructive Algebra[END_REF]. This paper can hence be seen as a generalization of the formalization of Gonthier to modules over rings instead over fields.

At the heart of the formalization of Gonthier is an implementation of Gaussian elimination which is used in all subspace constructions. Using it we can compute the kernel which characterizes the space of solutions of a system of linear equations. However when doing module theory over arbitrary rings, there is no general algorithm for solving systems of linear equations. Because of this we restrict our attention to modules over coherent and strongly discrete rings, as is customary in constructive algebra [START_REF] Lombardi | Algèbre commutative, Méthodes constructives: Modules projectifs de type fini[END_REF][START_REF] Mines | A Course in Constructive Algebra[END_REF], which means that we can solve systems of equations.

The main contributions of this paper are the representation of finitely presented modules over coherent strongly discrete rings (Sect. 2), basic operations on these modules (Sect. 3) and the formalization that the collection of these modules and morphisms forms an abelian category (Sect. 4), which means that it is a suitable setting for developing homological algebra. We have also proved that, over elementary divisor rings (i.e. rings with an algorithm to compute the Smith normal form of matrices), it is possible to test if two finitely presented modules represent isomorphic modules (Sect. 5). (Examples of such rings include principal ideal domains, in particular Z and k[X] where k is a field).

Finitely presented modules

As mentioned in the introduction, a module is finitely presented if it can be given by a finite set of generators and relations. This is traditionally expressed as:

Definition 1. An R-module M is finitely presented if there is an exact se- quence: R m1 R m0 M 0 M π
Recall that R m is the type of m-tuples of elements in R. More precisely, π is a surjection and M a matrix representing the m 1 relations among the m 0 generators of the module M. This means that M is the cokernel of M :

M coker(M ) = R m0 /im(M )
Hence a module has a finite presentation if it can be expressed as the cokernel of a matrix. As all information about a finitely presented module is contained in its presentation matrix we will omit the surjection π when giving presentations of modules.

Example 1. The Z-module Z ⊕ Z/2Z is given by the presentation:

Z Z 2 Z ⊕ Z/2Z 0 0 2 because if Z ⊕ Z/2Z is generated by (e 1 , e 2 ) there is one relation, namely 0e 1 + 2e 2 = 2e 2 = 0.
Operations on finitely presented modules can now be implemented by manipulating the presentation matrices, for instance if M and N are finitely presented R-modules given by presentations:

R m1 R m0 M 0 R n1 R n0 N 0 M N
the presentation of M ⊕ N is:

R m1+n1 R m0+n0 M ⊕ N 0   M 0 0 N  
We have represented finitely presented modules in Coq using the datastructure of matrices from the SSReflect library which is defined as:

Inductive matrix R m n := Matrix of {ffun 'I_m * 'I_n -> R}. (* With notations: *) (* 'M[R]_(m,n) = matrix R m n *) (* 'rV[R]_m = 'M[R]_(1,m) *) (* 'cV[R]_m = 'M[R]_(m,1) *)
where 'I_m is the type ordinal m which represents all natural numbers smaller than m. This type has exactly m inhabitants and can be coerced to the type of natural numbers, nat. Matrices are then represented as finite functions over the finite set of indices, which means that dependent types are used to express wellformedness. Finitely presented modules are now conveniently represented using a record containing a matrix and its dimension: The direct sum of two finitely presented modules is now straightforward to implement:

Definition dsum (M N : fpmodule R) := FPModule (block_mx (pres M) 0 0 (pres N)).

Here block_mx forms the block matrix consisting of the four submatrices. We now turn our attention to morphisms of finitely presented modules.

Morphisms

As for vector spaces we represent morphisms of finitely presented modules using matrices. The following lemma states how this can be done: Lemma 1. If M and N are finitely presented R-modules given by presentations:

R m1 R m0 M 0 R n1 R n0 N 0 M N
and ϕ : M → N a module morphism then there is a m 0 × n 0 matrix ϕ G and a m 1 × n 1 matrix ϕ R such that the following diagram commutes:

R m1 R m0 M 0 R n1 R n0 N 0 M ϕ R ϕ G ϕ N
For a proof of this see Lemma 2.1.25 in [START_REF] Greuel | A Singular Introduction to Commutative Algebra[END_REF]. This means that morphisms between finitely presented modules can be represented by pairs of matrices. The intuition why two matrices are needed is that the morphism affects both the generators and relations of the modules, hence the names ϕ G and ϕ R .

In order to be able to compute for example the kernel of a morphism of finitely presented modules we need to add some constraints on the ring R since, in general, there is no algorithm for solving systems of equations over arbitrary rings. The class of rings we consider are coherent and strongly discrete which means that it is possible to solve systems of equations. In Homalg these are called computable rings [START_REF] Barakat | An axiomatic setup for algorithmic homological algebra and an alternative approach to localization[END_REF] and form the basis of the system.

Coherent and strongly discrete rings

Given a ring R (in our setting commutative but it is possible to consider noncommutative rings as well [START_REF] Barakat | An axiomatic setup for algorithmic homological algebra and an alternative approach to localization[END_REF]) we want to study the problem of solving linear systems over R. If R is a field we have a nice description of the space of solutions by a basis of solutions. Over an arbitrary ring R there is in general no basis. For instance over the ring R = k[X, Y, Z] where k is a field, the equation pX + qY + rZ = 0 has no basis of solutions. It can be shown that a generating system of solutions is given by (-Y, X, 0), (Z, 0, -X), (0, -Z, Y ). An important weaker property than having a basis is that there is a finite number of solutions which generate all solutions. Definition 2. A ring is (left) coherent if for any matrix M it is possible to compute a matrix L such that:

XM = 0 ↔ ∃Y.X = Y L
This means that L generates the module of solutions of XM = 0, hence L is the kernel of M . For this it is enough to consider the case where M has only one column [START_REF] Lombardi | Algèbre commutative, Méthodes constructives: Modules projectifs de type fini[END_REF]. Note that the notion of coherent rings is not stressed in classical presentations of algebra since Noetherian rings are automatically coherent, but in a computationally meaningless way. It is however a fundamental notion, both conceptually [START_REF] Lombardi | Algèbre commutative, Méthodes constructives: Modules projectifs de type fini[END_REF][START_REF] Mines | A Course in Constructive Algebra[END_REF] and computationally [START_REF] Barakat | homalg -A Meta-Package for Homological Algebra[END_REF].

Coherent rings have previously been represented in Coq [START_REF] Coquand | Coherent and strongly discrete rings in type theory[END_REF], the only difference is that in the previous presentation, composition was read from right to left, whereas here we adopt the SSReflect convention that composition is read in diagrammatic order (i.e. left to right).

In the development, coherent rings have been implemented using the design pattern of [START_REF] Garillot | Packaging mathematical structures[END_REF], using packed classes and the canonical structure mechanism to help Coq automatically infer structures. As matrices are represented using dependent types denoting their size this needs to be known when defining coherent rings. In general the size of L cannot be predicted, so we include an extra function to compute this:

Record mixin_of (R : ringType) : Type := Mixin { dim_ker : forall m n, 'M[R]_(m,n) -> nat; ker : forall m n (M : 'M_(m,n)), 'M_(dim_ker M,m); _ : forall m n (M : 'M_(m,n)) (X : 'rV_m), reflect (exists Y, X = Y *m ker M) (X *m M == 0) }.
Here *m denotes matrix multiplication and == is the boolean equality of matrices, so the specification says that this equality is equivalent to the existence statement. An alternative to having a function computing the size would be to output a dependent pair but this has the undesirable behavior that the pair has to be destructed when stating lemmas about it, which in turn would make these lemmas cumbersome to use as it would not be possible to rewrite with them directly.

An algorithm that can be implemented using ker is the kernel modulo a set of relations, that is, computing ker(R m M -→ coker(N )). This is equivalent to computing an X such that ∃Y, XM + Y N = 0, which is the same as solving (X Y )(M N ) T = 0 and returning the part of the solution that corresponds to XM . In the paper this is written as ker N (M ) and in the formalization as N.ker(M). Note that this is a more fundamental operation than taking the kernel of a matrix as XM = 0 is also equivalent to ∃Y, X = Y ker 0 (M )

In order to conveniently represent morphisms we also need to be able to solve systems of the kind XM = B where B is not zero. In order to do this we need to introduce another class of rings that is important in constructive algebra: Definition 3. A ring R is strongly discrete if membership in finitely generated ideals is decidable and if x ∈ (a 1 , . . . , a n ) there is an algorithm computing w 1 , . . . , w n such that x = i a i w i .

Examples of such rings are multivariate polynomial rings over fields with decidable equality (via Gröbner bases) and Bézout domains (for instance Z and k[X] with k a field).

If a ring is both coherent and strongly discrete it is not only possible to solve homogeneous systems XM = 0 but also any system XM = B where B is an arbitrary matrix with the same number of columns as M . This operation can be seen as division of matrices as:

Lemma dvdmxP m n k (M : 'M[R]_(n,k)) (B : 'M[R]_(m,k)) : reflect (exists X, X *m M = B) (M %| B).
Here %| is notation for the function computing a particular solution to XM = B, returning None in the case no solution exists. We have developed a library of divisibility of matrices with lemmas like

Lemma dvdmxD m n k (M : 'M[R]_(m,n)) (N K : 'M[R]_(k,n)) : M %| N -> M %| K -> M %| N + K.
which follow directly from dvdmxP. This can now be used to represent morphisms of finitely presented modules and the division theory of matrices gives short and elegant proofs about operations on morphisms.

Finitely presented modules over coherent strongly discrete rings

Morphisms between finitely presented R-modules M and N can be represented by a pair of matrices. However when R is coherent and strongly discrete it suffices to only consider the ϕ G matrix as ϕ R can be computed by solving XN = M ϕ G , which is the same as testing N | M ϕ G . In Coq this means that morphisms between two finitely presented modules can be implemented as: Using this representation we can define the identity morphism (idm) and composition of morphisms (phi ** psi) and show that these form a category. We also define the zero morphism (0) between two finitely presented modules, the sum (phi + psi) of two morphisms and the negation (phi) of a morphism, respectively given by the zero matrix, the sum and the negation of the underlying matrices. It is straightforward to prove using the divisibility theory of matrices that this is a pre-additive category (i.e. the hom-sets form abelian groups).

However, morphisms are not uniquely represented by an element of type 'Mor(M,N), but it is possible to test if two morphisms ϕ ψ : M → N are equal by checking if ϕ -ψ is zero modulo the relations of N . As this is an equivalence relation it would be natural to either use the Coq setoid mechanism [START_REF] Barthe | Setoids in type theory[END_REF][START_REF] Sozeau | A new look at generalized rewriting in type theory[END_REF] or quotients [START_REF] Cohen | Pragmatic Quotient Types in Coq[END_REF] in order to avoid applying symmetry, transitivity and compatibility with operators (e.g. addition and multiplication) by hand where it would be more natural to use rewriting. We have begun to rewrite the library with quotients as we would get a set of morphisms (instead of a setoid), which is closer to the standard category theoretic notion.

Monomorphisms, epimorphisms and operations on morphisms

A monomorphism is a morphism ϕ : B → C such that whenever there are ψ 1 , ψ 2 : A → B with ψ 1 ϕ = ψ 2 ϕ then ψ 1 = ψ 2 . When working in pre-additive categories the condition can be simplified to, whenever ψϕ = 0 then ψ = 0.

Definition is_mono (M N : fpmodule R) (phi : 'Mor(M,N)) := forall (P : fpmodule R) (psi : 'Mor(P, M)), psi ** phi %= 0 -> psi %= 0.

It is convenient to think of monomorphisms B → C as defining B as a subobject of C, so a monomorphism ϕ : M → N can be thought of as a representation of a submodule M of N . However, submodules are not uniquely represented by monomorphisms even up to equality of morphisms (%=). Indeed, multiple monomorphisms with different sources can represent the same submodule. Although "representing the same submodule" is decidable in our theory, we chose not to introduce the notion of submodule, because it is not necessary to develop the theory.

Intuitively monomorphisms correspond to injective morphisms (i.e. with zero kernel). The dual notion to monomorphism is epimorphism, which intuitively corresponds to surjective morphism (i.e. with zero cokernel). For finitely presented modules, mono-(resp. epi-) morphisms coincide with injective (resp. surjective) morphisms, but this is not clear a priori. The goal of this section is to clarify this by defining when a finitely presented module is zero, showing how to define kernels and cokernels, and explicit the correspondence between injective (resp. surjective) morphisms and mono-(resp. epi-) morphisms.

Testing if finitely presented modules are zero

As a finitely presented module is the cokernel of a presentation matrix we have that if the presentation matrix of a module is the identity matrix of dimension n× n the module is isomorphic to n copies of the zero module. Now consider the following diagram:

R m0 R m0 0 m0 0 R m1 R m0 M 0 Im 0 X Im 0 M
which commutes if ∃X, XM = I m0 , i.e. when M | I m0 . Hence this gives a condition that can be tested in order to see if a module is zero or not.

Defining the kernel of a morphism

In order to compute the kernel of a morphism the key observation is that there is a commutative diagram: 0

R k1 R k0 ker(ϕ) 0 R m1 R m0 M 0 R n1 R n0 N 0 ker M (κ) X ker N (ϕ G )=κ M ϕ R ϕ G ϕ N
It is easy to see that κ is a monomorphism, which means that the kernel is a submodule of M as expected. In Coq this is easy to define: Where mor_of_mx takes a matrix K with as many columns as N and builds a morphism from ker N (K) to M . Using this it is possible to test if a morphism is injective:

Definition injm (M N : fpmodule R) (phi : 'Mor(M,N)) := kernel phi %= 0.

We have proved that a morphism is injective if and only if it is a monomorphism:

Lemma monoP (M N : fpmodule R) (phi : 'Mor(M,N)) : reflect (is_mono phi) (injm phi).

Hence we can define monomorphisms as: The reason why we use injm instead of is_mono is that injm is a boolean predicate, which makes monomorphisms a subtype of morphisms, thanks to Hedberg's theorem [START_REF] Hedberg | A Coherence Theorem for Martin-Löf's Type Theory[END_REF].

Defining the cokernel of a morphism

The presentation of the cokernel of a morphism can also be found by looking at a commutative diagram:

R m1 R m0 M 0 R n1 R n0 N 0 R m0+n1 R n0 coker(ϕ) 0 0 M ϕ R ϕ G ϕ N X In 0   ϕ G N  
Note that the canonical surjection onto the cokernel is given by the identity matrix. The fact that this is a morphism is clear as X may be 0 I n1 . However, before defining this we can define the more general operation of quotienting a module by the image of a morphism by stacking matrices:

Definition quot_by (M N : fpmodule R) (phi : 'Mor(M, N)) := FPModule (col_mx (pres N) phi)

Now the cokernel is the canonical surjection from N onto quot_by phi. Since it maps each generator to itself, the underlying matrix is the identity matrix. We can now test if a morphism is surjective by comparing the cokernel of phi with the zero morphism, which coincides with epimorphisms:

Definition surjm (M N : fpmodule R) (phi : 'Mor(M,N)) := coker phi %= 0.

Lemma epiP (M N : fpmodule R) (phi : 'Mor(M,N)) : reflect (is_epi phi) (surjm phi).

Now we have algorithms deciding both if a morphism is injective and surjective we can easily test if it is an isomorphism:

Definition isom (M N : fpmodule R) (phi : 'Mor(M,N)) := injm phi && surjm phi.

A natural question to ask is if we get an inverse from this notion of isomorphism. In order to show this we have introduced the notion of isomorphisms that take two morphisms and express that they are mutual inverse of each other, in the sense that given ϕ : M → N and ψ : N → M then ϕψ = 1 M modulo the relations in M . Using this we have proved:

Lemma isoP (M N : fpmodule R) (phi : 'Mor(M,N)) : reflect (exists psi, isomorphisms phi psi) (isom phi).

Hence isomorphisms are precisely the morphisms that are both mono and epi. Note that this does not mean that we can decide if two modules are isomorphic, what we can do is testing if a given morphism is an isomorphism or not.

Defining homology

The homology at N is defined as the quotient ker(ψ)/im(ϕ), in

M N K where ϕψ = 0. ϕ ψ
As ϕψ = 0, we have that im(ϕ) ⊂ ker(ψ) so the quotient makes sense and we have an injective map ι : im(ϕ) → ker(ψ). The homology at N is the cokernel of this map. We can hence write: Where %/ is a notation for taking the quotient of a monomorphism by a morphism with the same target.

In the next section, we show that these operations satisfy the axioms of abelian categories.

Abelian categories

As mentioned in the end of Sect. 2 the collection of morphisms between two finitely presented modules forms an abelian group. This means that the category of finitely presented modules and their morphisms is a pre-additive category. It is easy to show that the dsum construction provides both a product and coproduct. This means that the category is also additive.

In order to show that we have a pre-abelian category we need to show that morphisms have both a kernel and cokernel in the sense of category theory. A morphism ϕ : A → B has a kernel κ : K → A if κϕ = 0 and for all ψ : Z → A with ψϕ = 0 the following diagram commutes:

Z A B K 0 ψ ∃!ζ ϕ κ
This means that any morphism with ψϕ = 0 factors uniquely through the kernel κ. The dual statement for cokernels state that any morphism ψ with ϕψ = 0 factors uniquely through the cokernel of ϕ. The specification of the kernel can be written. We have proved that our definition of kernel satisfies this specification:

Lemma kernelP (M N : fpmodule R) (phi : 'Mor(M,N)) : is_kernel phi (kernel phi).

We have also proved the dual statement for cokernels. The only properties left in order to have an abelian category is that every mono-(resp. epi-) morphism is normal which means that it is the kernel (resp. cokernel) of some morphism. We have shown that if ϕ is a monomorphism then its cokernel satisfies the specification of kernels:

Lemma mono_ker (M N : fpmodule R) (phi : 'Mono(M,N)) : is_kernel (coker phi) phi.
This means that ϕ is a kernel of coker(ϕ) if ϕ is a monomorphism, hence are all monomorphisms normal. We have also proved the dual statement for epimorphisms which means that we indeed have an abelian category.

It is interesting to note that many presentations of abelian categories say that phi is kernel(coker phi), but this is not even well-typed as:

M N C K ϕ coker(ϕ)
ker(coker(ϕ))

One cannot just subtract ϕ and ker(coker(ϕ)) as they have different sources. This abuse of language is motivated by the fact that kernels are limits which are unique up to unique isomorphism which is why many authors speak of the kernel of a morphism. However, in order to express this formally we need to exhibit the isomorphism between M and K explicitly and insert it in the equation.

Note that if we introduced a notion of submodule, we could have defined the kernel as a unique submodule of N . The reason is that the type of submodules of N would be the quotient of monomorphisms into N by the equivalence relation which identifies them up to isomorphism.

Smith normal form

As mentioned before, it is in general not possible to decide if two presentations represent isomorphic modules, even when working over coherent strongly discrete rings. When the underlying ring is a field it is possible to represent a finite dimensional vector space in a canonical way as they are determined up to isomorphism by their dimension (i.e. the rank of the underlying matrix) which can be computed by Gaussian elimination [START_REF] Gonthier | Point-Free, Set-Free concrete linear algebra[END_REF]. A generalization of this is the class of rings, called elementary divisor rings by Kaplansky [START_REF] Kaplansky | Elementary divisors and modules[END_REF], where any matrix is equivalent to a matrix in Smith normal form. Recall that a matrix M is equivalent to a matrix D if there exist invertible matrices P and Q such that P M Q = D. Definition 4. A matrix is in Smith normal form if it is a diagonal matrix of the form:

           d 1 0 • • • • • • 0 . . . . . . 0 d k 0 • • • 0 . . . 0 0 . . . . . . . . . . . . . . . 0 • • • 0 • • • • • • 0           
where

d i | d i+1 for 1 i < k.
The connection between elementary divisor rings and finitely presented modules is that the existence of a Smith normal form for the presentation matrix gives us:

R m1 R m0 M 0 R m1 R m0 D 0 M P -1 Q ϕ D
Now ϕ is an isomorphism as P and Q are invertible. In order to represent this in Coq we need to represent diagonal matrices. For this we use the function diag_mx_seq. It is a function that takes two numbers m and n, a list s and returns a matrix of type 'M[R]_(m,n) where the elements of the diagonal are the elements of s. It is defined as follows:

Definition diag_mx_seq m n (s : seq R) := \matrix_(i < m, j < n) (s'_i *+ (i == j :> nat)).

This means that the i th diagonal element of the matrix is the i th element of the list and the rest are zero. Now if M is a matrix, our algorithm for computing the Smith normal form should return a list s and two matrices P and Q such that:

1. s is sorted by division and its length is less than m and n, 2. P *m M *m Q = diag_mx_seq m n s and 3. P and Q are invertible.

Any elementary divisor ring is coherent as the existence of an algorithm computing Smith normal form implies that we can compute kernels. Hence we only need to assume that R is a strongly discrete elementary divisor ring to be able to consider finitely presented modules over R. As P is invertible it is obvious that Q defines a morphism from M to diag_mx_seq m n s. Also P^-1 defines a morphism in the other direction that is inverse to P which means that M and diag_mx_seq m n s are isomorphic.

Bézout domains

We now assume that all rings have explicit divisibility, that is, we can decide if a | b and moreover produce x such that b = xa. Two elements a and b are associate if a | b and b | a. Since we are working over integral domains, a and b are associate if and only if there exists a unit u ∈ R such that a = bu. Definition 5. An integral domain R is a Bézout domain if every finitely generated ideal is principal (generated by a single element). This is equivalent to requiring that R has a GCD operation and a function computing the elements of the Bézout identity. This means that given a and b one can compute x and y such that xa + by and gcd(a, b) are associate.

We have formalized a proof that Bézout domains of Krull dimension less than or equal to 1 (in particular principal ideal domains like Z and k[X] with k a field) are elementary divisor rings, however as this paper is concerned with finitely presented modules we do not go into the details of this proof here. The reason why we restrict our attention to rings of Krull dimension less than or equal to 1 is that it is still an open problem whether all Bézout domains are elementary divisor rings or not [START_REF] Lorenzini | Elementary divisor domains and bézout domains[END_REF].

Combining this with finitely presented modules we get a constructive generalization to the classification theorem of finitely generated modules over principal ideal domains. This theorem states that any finitely presented R-module M over a principal ideal domain R can be decomposed into a direct sum of a free module and cyclic modules, that is, there exists n ∈ N and elements d 1 , . . . , d k ∈ R such that:

M R n ⊕ R/(d 1 ) ⊕ • • • ⊕ R/(d k )
with the additional property that d i | d i+1 for 1 i < k. In [START_REF] Cano | Matrices à blocs et en forme canonique[END_REF], it is formally proved in Coq that the Smith normal form is unique up to multiplication by units for rings with a GCD operation. This means that for any matrix M equivalent to a diagonal matrix D in Smith normal form, each of the diagonal elements of the Smith normal form of M will be associate to the corresponding diagonal element in D. This implies that the decomposition of finitely presented modules over elementary divisor rings is unique up to multiplication by units. This also gives a way for deciding if two finitely presented modules are isomorphic.

Conclusions and future work

In this paper we have presented a formalization of the category of finitely presented modules over coherent strongly discrete rings and shown that it is an abelian category. The fact that we can represent everything using matrices makes is possible for us to reuse basic results on these when building the abstraction layer of modules on top. The division theory of matrices over coherent strongly discrete rings makes it straightforward for us to do reasoning modulo a set of relations.

It is not only interesting that we have an abelian category because it provides us with a setting to do homological algebra, but also because it is proved in [START_REF] Coquand | Towards constructive homological algebra in type theory[END_REF] that in order to show that abelian groups (and hence the category of R-modules) form an abelian category in Coq one needs the principle of unique choice. As our formalization is based on the Mathematical Components hierarchy [START_REF] Garillot | Packaging mathematical structures[END_REF] of algebraic structures, we inherit a form of axiom of choice in the structure of discrete rings. However, we speculate that this axiom is in fact not necessary for our proof that the category of finitely presented modules over coherent strongly discrete rings is abelian.

In Homotopy Type Theory [START_REF]The Univalent Foundations Program: Homotopy Type Theory: Univalent Foundations of Mathematics[END_REF] there is a distinction between pre-categories and univalent categories (just called categories in [START_REF] Ahrens | Univalent categories and the Rezk completion[END_REF]). A pre-category is a category where the collection of morphisms forms a set in the sense of homotopy type theory, that is, they satisfy the uniqueness of identity proofs principle. Our category of finitely presented modules satisfy the uniqueness of morphism equivalence (phi %= psi) proofs (by Hedberg's theorem [START_REF] Hedberg | A Coherence Theorem for Martin-Löf's Type Theory[END_REF]), but morphisms form a setoid instead of a set. If we quotiented morphisms by the equivalence relation on morphisms we would get a set, and thus our category of finitely presented modules would become a pre-category.

A univalent category on the other hand is a pre-category where the equality of objects coincides with isomorphism. As we have shown that for elementary divisor rings there is a way to decide isomorphism, we speculate that we would also get a univalent category by quotienting modules by isomorphisms. It would be interesting to develop these ideas further and define the notion of univalent abelian category and study its properties. Note that in Homotopy Type Theory, it may be no longer necessary to have the decidability of the equivalence relation to form the quotient, so we would not need to be in an elementary divisor ring to get a univalent category.

Since we have shown that we have an abelian category it would now be very interesting to formally study more complex constructions from homological algebra. It would for instance be straightforward to define resolutions of modules. We can then define define the Hom and tensor functors in order to get derived functors like T or and Ext. It would also be interesting to define graded objects like chain complexes and graded finitely presented modules, and prove that they also form abelian categories.

  Record fpmodule := FPModule { nbrel : nat; nbgen : nat; pres : 'M[R]_(nbrel, nbgen) }.

  Record morphism_of (M N : fpmodule R) := Morphism { matrix_of_morphism : 'M[R]_(nbgen M,nbgen N); _ : pres N %| pres M *m matrix_of_morphism }. (* With notation: *) (* 'Mor(M,N) := morphism_of M N *)

  Definition eqmor (M N : fpmodule R) (phi psi : 'Mor(M,N)) := pres N %| phi%:m -psi%:m. (* With notation: *) (* phi %= psi = eqmor phi psi *)

  Definition kernel (M N : fpmodule R) (phi : 'Mor(M,N)) := mor_of_mx ((pres N).-ker phi).

  Record monomorphism_of (M N : fpmodule R) := Monomorphism { morphism_of_mono :> 'Mor(M, N); _ : injm morphism_of_mono }. (* With notation: *) (* 'Mono(M,N) = monomorphism_of M N *)

  Definition coker : 'Mor(N, quot_by) := Morphism1 (dvd_quot_mx (dvdmx_refl _)).

  Hypothesis mul_phi_psi (M N K : fpmodule R) (phi : 'Mor(M,N))(psi : 'Mor(N,K)) : phi ** psi %= 0.Definition homology (M N K : fpmodule R) (phi : 'Mor(M,N)) (psi : 'Mor(N,K)) := kernel psi %/ phi.

  Definition is_kernel (M N K : fpmodule R) (phi : 'Mor(M,N)) (k : 'Mor(K,M)) := (k ** phi %= 0) * forall L (psi : 'Mor(L,M)), reflect (exists Y, Y ** k %= psi) (psi ** phi %= 0).

The formal development is at: http://perso.crans.org/cohen/work/fpmods/

